
Static analysis approximates the results of a program by examining only its syntax. For example, control-flow
analysis (CFA) determines which syntactic lambdas (for functional languages) or (for object-oriented) methods
may be invoked at each call site within a program. Rich theoretical results exist studying control flow analysis
for Scheme-like languages, but implementations are often complex and specialized. By contrast, object-oriented
languages (Java in particular) enjoy high-precision control-flow analyses that scale to thousands (or more) of
lines of code. State-of-the-art implementations (such as DOOP on Soufflé) structure the analysis using Horn-
SAT (Datalog) to enable compilation of the analysis to efficient implementations such as high-performance
relational algebra kernels. In this paper, we present an implementation of control-flow analysis for a significant
subset of Scheme (including set!, call/cc, and primitive operations) using the Soufflé Datalog engine. We
present an evaluation on a worst-case term demonstrating the polynomial complexity of our𝑚-CFA and
remark upon scalability results using Soufflé.

1 INTRODUCTION
Static analysis is a technique to explicate properties of a program’s behavior via inspecting only
the program’s source code (without executing the program) [13]. There exist many frameworks
for constructing program analyses, e.g., Cousot and Cousot’s abstract interpretation [5]. A static
analysis is sound when it is strictly conservative in the sense that any true program behavior is
reported (at least approximately) as a result of the analysis. Unfortunately—due to the halting
problem—no terminating static analysis may be both sound and complete (all reported results
represent true behavior). While static analyses may be constructed using arbitrary degrees of
precision (e.g., via instrumentation-based polyvariance [7]) in principle, in practice balancing
precision and complexity while retaining soundness requires significant engineering effort [2].

A central challenge in analyzing Scheme programs is control-flow analysis: for each callsite,
which syntactic lambdas (in which contexts, for a context-sensitive analysis) may be invoked? This
problem is simple in procedural languages (which include only direct control-flow), but challenging
for higher-order languages, as data-flow and control-flow must be performed simultaneously.
Shivers defined the 𝑘-CFA family of increasingly-precise control-flow analyses for Scheme [16].
However, as Shivers notes, uniform 𝑘-CFA quickly becomes intractable for the case 𝑘 > 0, even
for reasonably-sized programs [15]. In fact, Van Horn and Mairson later showed that 𝑘-CFA is
EXPTIME-complete [18].
Compared to Scheme, significant engineering work has been expended into whole-program

analysis for object-oriented languages, particularly Java [1, 3, 9, 14]. This has culminated in state-of-
the-art systems such as DOOP, whose analysis is written in Datalog for subsequent compilation to
efficient relational algebra kernels implemented via C++ [3]. Systems such as DOOP scale to large
(multi-thousand line) codebases even when context-sensitivity is considered. This would appear at
odds with Van Horn and Mairson’s result that 𝑘-CFA is EXPTIME-complete. To resolve this paradox,
Might et al. show that the degenerative closure structure of object-oriented languages (analogous
to the difference between flat and linked closures), context-sensitive control-flow analyses of object-
oriented languages is PTIME [11]. The authors also present𝑚-CFA, a technique to analyze arbitrary
higher-order languages using flat closures to achieve polynomial complexity.

While the techniques for analyzing object-oriented languages using Datalog are well understood,
we are aware of no existing presentation illustrating how Datalog may be used to implement the
analysis of Scheme. We find this particularly surprising, as we believe the extreme efficiency of
modern Datalog engines may prove a key enabling technology to tackle the inherent complexity of
higher-order languages.
In this paper, we present a systematic approach for deriving a Datalog-based implementation

of control-flow analysis for Scheme-like languages. Key to our approach is the formulation of the
analysis via the abstracting abstract machines (AAM) methodology of Van Horn and Might [19].
We present how this formulation may be translated to Datalog, overcoming several key obstacles
unique to Scheme. To evaluate our approach, we implemented an analysis of a significant subset
of Scheme including multi-argument lambdas, conditional control-flow, builtins, and first-class
continuations (call/cc). In section 3, we present a formalization of this language (noting how our
choices anticipate a Datalog implementation) via the AAM approach, following in section 4 with
its corresponding Datalog transliteration. We validated (manually, via inspection) the correctness
of our analysis and in section 5 present the results of a set of experiments benchmarking our
implementation of𝑚-CFA for our subset of Scheme.

2 BACKGROUND AND RELATEDWORK
In this section, we sketch several key background concepts that underlie our abstract semantics in
Section 3 and subsequent Datalog implementation in Section 4.

2.1 Program Analysis and Abstract Interpretation
Kildall first introduced dataflow analysis (of flowchart-style programs in the style of Floyd [6])
to approximate static program behavior for the purpose of compile-time optimization [10]. A
central idea of Kildall’s work was using a lattice to impose an ordering on analysis results and
ensure termination via the finiteness of said lattice. Cousot and Cousot later generalized Kildall’s
ideas to define the abstract interpretation of a program. Abstract interpretation allows relating
an arbitrary pair of lattices, typically a concrete state space (Σ) and its abstraction (Σ̂), along with
a pair, (𝛼,𝛾), of (adjunctive) mappings between them for abstraction (𝛼) and concretization (𝛾 ).
Using this abstraction alongside a collecting semantics allows iterating a program analysis to some

fixed-point in an arbitrary lattice of abstract results. Assuming this lattice of results is of finite
height, the collecting semantics (and therefore analysis) will necessarily terminate via Tarski’s
fixpoint theorem [17].

We elide a complete presentation of abstract interpretation; there exist several expository texts
including those by Miné [12] and Nielsen and Nielsen [13].

2.2 Control Flow Analysis
Languages such as C do not include indirect control flow. Determining control flow (and also
data flow) is simple for these languages, as control is syntactically apparent. In Scheme, it is more
difficult to tell which values flow to a particular variable because of the pervasive use of higher-order
functions.

Consider the following Scheme code:
(let* ([f (foo 42)]

[g (bar 99)]
[h (if (= a b) (g 30) (f g))])

(g h))

Deciding which branch of the if is taken depends on (at least) the values that flow to a and b.
Similarly, to understand data flow, we also must know control flow: knowing which value flows
to h requires reasoning about the if’s control flow. The key is to compute both simultaneously.
As the computation continues, data flow information is fed to create a control-flow graph (CFG)
on-the-fly, and the new CFG is used to find new data-flow. This is a central idea in the original
formulation of 𝑘-CFA by Shivers [16], though presentations can also be found elsewhere [12, 13].

2.3 Abstract Abstract Machines
Might and Van Horn presented the Abstracting Abstract Machines (AAM) approach to abstract
interpretation for functional languages [19]. The key insight in their work is to redirect all sources
of recursion in the analysis through a store which may be finitized by construction. Using this
approach, an abstract semantics may be derived from an abstract machine specifying a concrete
semantics. The AAM-based approach can compute any type of CFA, and encompasses a broad
array of analysis precision including, e.g., object-sensitivity [7]. The machine described in this
paper utilizes𝑚-CFA [11], a variant of 𝑘-CFA that uses flat closures.

2.4 Datalog
Datalog is a bottom-up logic programming language largely based on Horn-SAT. We refer the
reader to the exposition of Ceri et al. for a detailed description of Datalog [4]. Datalog programs
consist of a set of Horn clauses of the form 𝑃 (𝑥0, ...) ← 𝑄 (𝑦0, ...) ∧ ... ∧ 𝑆 (𝑧0, ...). To evaluate these
programs, an extensional database (EDB) is provided as input specifying a set of initial facts. A
Datalog engine then runs the rules to a fixed-point to produce an output database. The following
example computes the cousin relation from the EDB relations of parent and sibling.
cousin(a, c) :- parent(a, p), sibling(p, q), parent(c, q).

Relations can be recursive. Calculating ancestry is simple. The base case shows that a parent is
trivially an ancestor, but a parent of an ancestor is also an ancestor:
ancestor(p, a) :- parent(p, a).
// If p already has some ancestor a,
// the parent b of a is also an ancestor of p.
ancestor(p, b) :- ancestor(p, a), parent(a, b).

3 SYNTAX AND ABSTRACT SEMANTICS

Syntactic Classes

𝑒 ∈ Exp ::= æ
| (if 𝑒 𝑒 𝑒) | (set! 𝑥 𝑒)
| (call/cc 𝑒) | let
| (𝑜𝑝 𝑒 𝑒) | (𝑒 𝑒 𝑒 ...)

æ ∈ AExp ::= 𝑥 | lam | 𝑏 | 𝑛

𝑏 ∈ B ≜ {#t, #f}
𝑛 ∈ Z

𝑥 ∈ Var ≜ The set of identifiers
let ∈ Let ::= (let ((𝑥 𝑒) ...) 𝑒)

lam ∈ Lam ::= (_ (𝑥) 𝑒)
op ∈ Prim ≜ The set of primitives

Fig. 1. Syntactic Classes for the Scheme CESK* Machine

Many famous papers use a significant subset of Scheme, perhaps too small for useful real-world
analyses. Might et al.’s m-CFA paper uses a variant of the CPS lambda calculus with multi-argument
functions [11]. Shivers’ original work on CFA utilized a CPS subset without direct conditionals
or mutation [16]. Van Horn et al.’s original Abstracting Abstract Machines adds direct mutation,
conditionals, and first order continuations, but does not include multi-argument lambdas, which is
a crucial component of CFA [19], as they preserve contexts more thoughtfully. Each of these leaves
out important components which are required to build analyses on real languages.
In Figure 1, we partition complex and atomic expressions. We also include a variety of useful

syntax such as mutation through set! expressions, let bindings, and the higher order control flow
operator, call/cc. Conditional expressions and primitives are also highly important for writing
meaningful examples. We support binary primitives, which are needed to analyze real Scheme
programs. Primitive ops include constructing lists and mathematical and logical operations.
The subset of Scheme used is capable of supporting many real world programs. Results of

analyses using this subset are more clear than an analysis of a heavily compiled subset. Results of
analyses are more understandable and easier to implement due to these features. This syntax also
clarifies contexts in a CFA and how they grow and shrink. With let and multi-argument lambdas,
we can place multiple bindings in a single context. This will avoid states trivially ascending to top,
by decreasing the number of contexts.
Figure 2 presents a CESK* machine utilizing𝑚-CFA for value analysis, [11, 19]. The states are

partitioned into evaluation and application states, and the store is partitioned into a value store and
a continuation store. There are a variety of continuations which give meaning to the program.
The set of states is partitioned into 2 types of sub-state. The control of an Eval state is syntax.

When transitioning from an Eval state, the goal is to produce a value. For example, when an if
expression is encountered, the resulting state is another Eval state with the guard expression set to
the control. The second type of state is an Apply, where the control is a value. These states will
apply the control depending on the continuation. When an if expression’s guard reaches an Apply
state, a branch is selected based on the inspected value.

With abstracted abstract machines, both values and continuations are identified through the store
with an address. However, they require different address types. Value addresses are determined by
the polyvariance and analyses type we are conducting [7]. Continuations, however, are allocated
in the Pushdown-For-Free style, which has its own allocator [8]. These stores are combined in the
state, but are accessed separately and combined when clear in the semantics for brevity.
Environments in this machine are not a mapping, as shown in many AAM based approaches.

Instead, environments are simply a function of context, and stand-in for time-stamps as in 𝑘-CFA

Semantic Classes

𝜍 ∈ Σ̂ ≜ 𝐸⟨Êval⟩ +𝐴⟨�Apply⟩
Êval ≜ Exp ×�Context
×�Store ×�KAddr�Apply ≜ V̂al ×�Store ×�KAddr

�̂� ∈�Store ≜ �VStore ×�KStore
�̂�𝑣 ∈�VStore ≜�VAddr ⇀ P(V̂al)
�̂�𝑘 ∈ �KStore ≜ �KAddr ⇀ P(K̂ont)

𝑐𝑡𝑥 ∈�Context ≜ Exp𝑚

𝑎𝑣 ∈�VAddr ≜ Var ×�Context
𝑎 ˆ̂ ∈ �KAddr ≜ Exp ×�Context

𝑣 ∈ V̂al ≜ Z + B + P̂Val + Ĉlo +�KAddr
p̂ ∈ P̂Val ≜ Prim × V̂al × V̂al

𝑐𝑙𝑜 ∈ Ĉlo ≜ Lam ×�Context
ˆ̂ ∈ K̂ont ::= m̂tk | îfk(𝑒, 𝑒, 𝑐𝑡𝑥, 𝑎 ˆ̂ )

| ŝetk(𝑎𝑣, 𝑎 ˆ̂ ) | �callcck(𝑐𝑡𝑥, 𝑎 ˆ̂ )

| l̂et(𝑎𝑣, 𝑒, 𝑐𝑡𝑥, 𝑎 ˆ̂ ) | f̂n(𝑣, 𝑛, 𝑐𝑡𝑥, 𝑎 ˆ̂ )
| p̂1(op, 𝑒, 𝑐𝑡𝑥, 𝑎 ˆ̂ ) | p̂2(op, 𝑣, 𝑎 ˆ̂ )
| ârg(𝑒 ..., 𝑐𝑡𝑥, 𝑐𝑡𝑥, 𝑎 ˆ̂ )

Fig. 2. Semantic Classes for the Scheme m-CFA CESK* Machine

[11]. Here, closures utilize flat-contexts, instead of the usual linked model. Instead of store-fetches
being �̂� (𝑐𝑡𝑥 (𝑥)), they are now �̂� (𝑥, 𝑐𝑡𝑥). In𝑚-CFA, the current context is based on the top𝑚 stack
frames, as opposed to the latest 𝑘 call-sites.
There are a variety of continuation types to enumerate the various semantic features of the

Scheme subset. When a set! expression is evaluated, a ŝetk continuation is added to the store
and pushed onto the stack. This identifies what to accomplish after the inner expression is fully
evaluated.
The Eval rules in Figure 3 govern how to break down complex expressions to form an atomic

expression. Some rules are straightforward, such as E-If. After encountering an if expression, the
guard is evaluated. The îfk continuation is created to keep track of what to do when we know the
value of the guard. Generally, an evaluation rule focuses on evaluating syntax, and not creating
semantic objects such as contexts.
However other transition rules are trickier. The E-Let rule transitions to multiple states, one

for each binding. For brevity, at least one binding is required, as a 0-binding let would have to
transition directly to the body with a new context. Let is also an interesting case since it creates
the resulting context during syntactic evaluation. Enough information is known at evaluation time
to create it, so it is done then. The later application rule can then be more simple.
The E-Prim and E-Call rules are multi-stage rules. They each have multiple expressions that

must be evaluated in a specific order. In a function call, the function is evaluated before the
arguments, so there must be a special continuation frame for each juncture in the evaluation.

Atomic evaluation is defined in Figure 4. The function Â produces values from atomic expressions.
The rule E-AE will transition from an Eval state into an Apply state using Â.

Apply rules, as shown in Figure 5, transition based on the value and the current continuation.
These rules primarily govern contexts. Here, the context may be extended, bindings added to it, or
it may be returned to a previous version.

Evaluation Rules
𝐸⟨(if 𝑒𝑔 𝑒𝑡 𝑒𝑓 ), 𝑐𝑡𝑥, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑔, 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩

where 𝑎′ˆ̂ ≜�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, 𝑒𝑐 , 𝑐𝑡𝑥) (E-If)

ˆ̂ ≜ îfk(𝑒𝑡 , 𝑒𝑓 , 𝑐𝑡𝑥, 𝑎 ˆ̂ )
�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′ˆ̂ ↦→ ˆ̂]

𝐸⟨(op 𝑒0 𝑒1), 𝑐𝑡𝑥, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒0, 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩
where 𝑎′ˆ̂ ≜�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, 𝑒𝑖 , 𝑐𝑡𝑥) (E-Prim)

ˆ̂ ≜ p̂1(op, 𝑒1, 𝑐𝑡𝑥, 𝑎 ˆ̂ )
�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′ˆ̂ ↦→ ˆ̂]

𝐸⟨let, 𝑐𝑡𝑥, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑖 , 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩
where 𝑐𝑡𝑥 ′ ≜ 𝑛𝑒𝑤 (𝜍) (E-Let)

let = (let ((𝑥0 𝑒0) (𝑥𝑠 𝑒𝑠 ) ...) 𝑒𝑏)
(𝑥𝑖 , 𝑒𝑖 ) ∈ ([𝑥0 : 𝑥𝑠 ], [𝑒0 : 𝑒𝑠 ])

𝑎𝑣 ≜ �𝑎𝑙𝑙𝑜𝑐𝑣 (𝑥𝑖 , 𝜍)
𝑎′ˆ̂ ≜

�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, 𝑒𝑖 , 𝑐𝑡𝑥)
ˆ̂ ≜ l̂et(𝑒𝑏, 𝑎𝑣, 𝑐𝑡𝑥

′
, 𝑎 ˆ̂ )

�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′ˆ̂ ↦→ ˆ̂]

𝐸⟨(call/cc 𝑒), 𝑐𝑡𝑥, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒, 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩
where 𝑐𝑡𝑥 ′ ≜ 𝑛𝑒𝑤 (𝜍) (E-C/cc)

𝑎′ˆ̂ ≜
�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, 𝑒, 𝑐𝑡𝑥)

ˆ̂ ≜ �callcck(𝑐𝑡𝑥 ′, 𝑎 ˆ̂ )
�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′ˆ̂ ↦→ ˆ̂]

𝐸⟨(set! 𝑥 𝑒), 𝑐𝑡𝑥, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒, 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩
where 𝑎′ˆ̂ ≜�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, (set! 𝑥 𝑒), 𝑐𝑡𝑥) (E-Set!)

𝑎𝑣 ≜ �𝑎𝑙𝑙𝑜𝑐𝑣 ((set! 𝑥 𝑒), 𝜍)

ˆ̂ ≜ ŝetk(𝑎𝑣, 𝑎 ˆ̂ )
�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′ˆ̂ ↦→ ˆ̂]

𝐸⟨(𝑒𝑓 𝑒0 𝑒𝑠 ...), 𝑐𝑡𝑥, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑓 , 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩
where 𝑐𝑡𝑥 ′ ≜ 𝑛𝑒𝑤 (𝜍) (E-Call)

𝑎′ˆ̂ ≜
�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, (𝑒𝑓 𝑒0 𝑒𝑠 ...), 𝑐𝑡𝑥 ′, 𝑎 ˆ̂ )

ˆ̂ ≜ ârg( [𝑒0 : 𝑒𝑠 ], 𝑐𝑡𝑥, 𝑐𝑡𝑥
′
, ˆ̂)

�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′ˆ̂ ↦→ ˆ̂]

Fig. 3. Rules to evaluate syntax into smaller expressions

Atomic Evaluation

𝐸⟨æ, _, �̂�, 𝑎 ˆ̂ ⟩ { 𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩
where 𝑣 ≜ Â(𝜍) (E-AE)

Â :: Êval→ 𝑣

Â(𝑏, _, _) ≜ 𝑏 Â(𝑛, _, _) ≜ 𝑛

Â(𝑥, 𝑐𝑡𝑥, _) △∋ �̂�𝑣 (𝑥, 𝑐𝑡𝑥)

Â(𝑙𝑎𝑚, 𝑐𝑡𝑥, _) ≜ (𝑙𝑎𝑚, 𝑐𝑡𝑥)

Fig. 4. Atomic Evaluation, which converts syntax into a value.

Apply rules may be triggered at the same time. For example, if an address contains multiple
values, then both A-If rules must be triggered. In Figure 6, both branches must be taken to soundly
approximate this program. In the Figure 6 example, the inner let calls the same function twice,
which, in the same context, binds the address 𝑥 to both #t and #f. Then, when 𝑎 is referenced, it
contains both possibilities, so both branches will be taken.

The A-Let rule will be applied once for every binding in a let expression. Because the addresses
were calculated in the E-Let rule, not much work needs to be done in this rule. However, because

Application Rules
𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑡 , 𝑐𝑡𝑥, �̂�, 𝑎′ˆ̂ ⟩

where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ îfk(𝑒𝑡 , _, 𝑐𝑡𝑥, 𝑎′ˆ̂ ) (A-IfT)
𝑣 ≠ #f

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑓 , 𝑐𝑡𝑥, �̂�, 𝑎′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ îfk(_, 𝑒𝑓 , 𝑐𝑡𝑥, 𝑎′ˆ̂ ) (A-IfF)

𝑣 = #f

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑏, 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩

where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ l̂et(𝑒𝑏, 𝑎𝑣, 𝑐𝑡𝑥, 𝑎′ˆ̂ ) (A-Let)
�̂� ′𝑣 ≜ �̂�𝑣 ⊔ [𝑎𝑣 ↦→ 𝑣]

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐴⟨−42, �̂�, 𝑎′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ ŝetk(𝑎𝑣, 𝑎′ˆ̂ ) (A-Set!)

�̂� ′𝑣 ≜ �̂�𝑣 ⊔ [𝑎𝑣 ↦→ 𝑣]

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒, 𝑐𝑡𝑥, �̂� ′, 𝑎′′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ p̂1(op, 𝑒, 𝑐𝑡𝑥, 𝑎′ˆ̂ ) (A-Prim1)

𝑎′′ˆ̂ ≜
�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, 𝑒, 𝑐𝑡𝑥)

ˆ̂ ≜ p̂2(op, 𝑣, 𝑎′ˆ̂ )
�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′′ˆ̂ ↦→ ˆ̂]

𝐴⟨𝑣 ′, �̂�, 𝑎 ˆ̂ ⟩ { 𝐴⟨primVal(op, 𝑣, 𝑣 ′), �̂�, 𝑎′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ p̂2(op, 𝑣, 𝑎′ˆ̂ ) (A-Prim2)

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑏, 𝑐𝑡𝑥, �̂� ′′, 𝑎′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ �callcck(𝑐𝑡𝑥, 𝑎′ˆ̂ ) (A-C/cc)

𝑣 = ((_ (𝑥) 𝑒𝑏), 𝑐𝑡𝑥𝑐𝑙𝑜 )

𝑎𝑣 ≜ �𝑎𝑙𝑙𝑜𝑐𝑣 (𝑥, 𝜍)
�̂� ′𝑣 ≜ ĉopy(𝑐𝑡𝑥

𝑐𝑙𝑜
, 𝑐𝑡𝑥)

�̂� ′′𝑣 ≜ �̂�𝑣 ⊔ [𝑎𝑣 ↦→ 𝑎 ˆ̂ ]

𝐴⟨𝑎 ˆ̂ , �̂�, 𝑎
′
ˆ̂ ⟩ { 𝐴⟨𝑎′ˆ̂ , �̂�, 𝑎 ˆ̂ ⟩

where �̂� ˆ̂ (𝑎′ˆ̂ ) ∋ �callcck(_, _) (A-C/ccKont)

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑖 , 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ ârg(𝑒𝑠 , 𝑐𝑡𝑥, 𝑐𝑡𝑥

′
, 𝑎′ˆ̂ ) (A-Ar)

𝑒𝑖 ∈ 𝑒𝑠
𝑎′ˆ̂ ≜

�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, 𝑒𝑖 , 𝑐𝑡𝑥)
ˆ̂ ≜ f̂n(𝑣, 𝑖, 𝑐𝑡𝑥 ′, 𝑎′ˆ̂ )

�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎𝑣 ↦→ ˆ̂]

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑏, 𝑐𝑡𝑥, �̂� ′′, 𝑎′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ f̂n(𝑐𝑙𝑜, 𝑛, 𝑐𝑡𝑥, 𝑎′ˆ̂ ) (A-Call)

𝑐𝑙𝑜 = ((_ (𝑥𝑠 ...) 𝑒𝑏), 𝑐𝑡𝑥𝑐𝑙𝑜 )

𝑎𝑣 ≜ �𝑎𝑙𝑙𝑜𝑐𝑣 (𝑥𝑛, 𝜍)
�̂� ′𝑣 ≜ ĉopy(𝑐𝑡𝑥

𝑐𝑙𝑜
, 𝑐𝑡𝑥)

�̂� ′′𝑣 ≜ �̂�𝑣 ⊔ [𝑎𝑣 ↦→ 𝑣]

𝐴⟨𝑣, �̂�, 𝑎 ˆ̂ ⟩ { 𝐴⟨𝑣, �̂�, 𝑎′ˆ̂ ⟩
where �̂� ˆ̂ (𝑎 ˆ̂ ) ∋ f̂n(𝑎′ˆ̂ , _, _, _) (A-CallKont)

Fig. 5. Value Application Rules

(let ([f (lambda (x) x)])
(let ([a (f #t)] [b (f #f)])
(if a 4 5)))

Fig. 6. An example where both branches of a condition must be taken.

every l̂et continuation results in the same state, only one output state is generated for the body of
the expression.

Creating new contexts is only done when variables are bound. Therefore, we only need to create
a context in the E-Let, E-C/cc, and E-Call rules. The new context is used to evaluate the expression
containing the new bindings. Therefore, the new context only transitions in the corresponding
Apply rule.

Helper Functions

Î :: Exp→ Σ̂

Î (𝑒) ≜ 𝐸⟨𝑒, 𝜖, (⊥, �̂� ˆ̂ ), (𝑒, 𝜖)⟩
�̂� ˆ̂ ≜ ⊥ ⊔ [(𝑒, 𝜖) ↦→ m̂tk]

𝑛𝑒𝑤 :: Êval→�Context
𝑛𝑒𝑤 (𝑒, 𝑐𝑡𝑥, _) ≜ ⌊𝑒 : 𝑐𝑜𝑛𝑡𝑒𝑥𝑡⌋𝑚

�𝑎𝑙𝑙𝑜𝑐𝑣 :: Var × Σ̂ ⇀ �VAddr�𝑎𝑙𝑙𝑜𝑐𝑣 (𝑥, 𝐸⟨_, 𝑐𝑡𝑥, _, )⟩ ≜ (𝑥, 𝑐𝑡𝑥)�𝑎𝑙𝑙𝑜𝑐𝑣 (𝑥,𝐴⟨(_, 𝑐𝑡𝑥), _, )⟩ ≜ (𝑥, 𝑐𝑡𝑥)�𝑎𝑙𝑙𝑜𝑐𝑘 :: Σ̂ × Exp ×�Context ⇀ �KAddr�𝑎𝑙𝑙𝑜𝑐𝑘 (_, 𝑒, 𝑐𝑡𝑥) ≜ (𝑒, 𝑐𝑡𝑥)
Fig. 7. Auxiliary functions

Transition Relation and Global Store

({) :: Σ̂ ⇀ P(Σ̂)
({Ξ̂) :: Ξ̂ ⇀ P(Ξ̂)
(𝑟, �̂�) {Ξ̂ (𝑟

′, �̂� ′)
where 𝑠 ≜ {𝜍 | (𝑟, �̂�) { 𝜍} ∪ {Î(𝑒𝑜 )

𝑟 ′ ≜ {𝑟 | (𝑟, _) ∈ 𝑠}

�̂� ′ ≜
⊔
(_,�̂�′′) ∈𝑠

�̂� ′′

b̂ ∈ Ξ̂ ≜ 𝑅 ×�Store
𝑟 ∈ 𝑅 ≜ P(𝐶)

𝑐 ∈ 𝐶 ≜ 𝐶𝐸⟨Exp ×�Context ×�KAddr⟩
+𝐶𝐴⟨V̂al ×�KAddr⟩

Fig. 8. Transition Relation and Global Store

The value and continuation stores are necessarily responsible for an exponential growth of states
in the state space. One method to maintain polynomial worst-case complexity when evaluating
closures is to separate the stores from the states, and to globalize the stores. After each state
transition, the stores are combined and used for the next transitions. Figure 8 gives a collecting
semantics which transforms a standard machine with an inline store into a global store. The
semantics utilize the injection function Î, along with the inline store transition function{.

4 DATALOG IMPLEMENTATION
Soufflé is used as the Datalog engine for its state-of-the-art runtime performance, parallelism, and
language features. Algebraic Data Types (ADT) are utilized for contexts, value and continuation
types to maintain brevity. This implementation can be replicated without ADTs but it will be much
more verbose.

Implementing the above operational semantics as a Datalog program is a straightforward process.
Many rules can be converted in a near 1:1 fashion. However, there are some unique relations
that differ from the operational semantics. Still, the rules of the machine map closely to the
implementation in Datalog. Figure 9 shows the similarity between the two.

Datalog programs being split into rules maps well onto operational semantics. There is a corre-
spondence between the operational semantics and the Datalog implementation. Datalog places the

state_e(eguard, ctx, ak),
stored_kont(ak, kont),
flow_ee(e, eguard) :-

state_e(e, ctx, ak),
if(e, eguard, et, ef),
ak = $KAddress(eguard, ctx),
kont = $If(et, ef, ctx, ak).

𝐸⟨(if 𝑒𝑔 𝑒𝑡 𝑒𝑓 ), 𝑐𝑡𝑥, �̂�, 𝑎 ˆ̂ ⟩ { 𝐸⟨𝑒𝑔, 𝑐𝑡𝑥, �̂� ′, 𝑎′ˆ̂ ⟩
where 𝑎′ˆ̂ ≜�𝑎𝑙𝑙𝑜𝑐𝑘 (𝜍, 𝑒𝑐 , 𝑐𝑡𝑥) (E-If)

ˆ̂ ≜ îfk(𝑒𝑡 , 𝑒𝑓 , 𝑐𝑡𝑥, 𝑎 ˆ̂ )
�̂� ′ˆ̂ ≜ �̂� ˆ̂ ⊔ [𝑎′ˆ̂ ↦→ ˆ̂]

Fig. 9. If expression evaluation in Datalog and in the operational semantics.

peek_ctx(e, old_ctx, new_ctx) :-
state_e(e, old_ctx, _),
( callcc(e, _) ; call(e, _, _)
; let(e, _, _); lambda(e,_,_)),
old_ctx = $Context(ctx1,ctx0),
new_ctx = $Context(e,ctx1).

stored_val(av, v) :-
copy_ctx(from, to, e),
freevar(fv, e),
stored_val(av, v),
av = $VAddress(fv, to).

Fig. 10. The new and copy function analogues in the Datalog implementation.

head of the Horn clause before the body, so the output state is on top, along with the continuation
being added to the store. The body of the Datalog rule acts as the input. As rules are computed, new
facts are generated by the heads of the clauses. As a result, more rules can be executed, extending
the set of total facts.
Figure 10 shows two interesting relations. These highlight a key difference between the opera-

tional semantics and the Datalog implementation. On the left is the equivalent to 𝑛𝑒𝑤 : peek_ctx.
On the right is the implementation of ĉopy. In Datalog, we compute a helper relation, peek_ctx,
which computes a context as needed. The copy_ctx relation will signal a copy operation, from a
source to a destination context. When a copy_ctx fact is added, it does not immediately do the
copy, but it will trigger an inference rule for the stored_val relation.

See Appendix A for the full Datalog implementation.

5 EVALUATION
In this section we present an evaluation of our implementation of 𝑚-CFA in Soufflé. To gain
confidence in our implementation’s correctness, we performed manual validation on a set of
testcases. To measure scalability and complexity in practice, we construct a family of terms that
exhibit worst-case (polynomial) behavior. We used large terms in this family and ran experiments
on a 28-core Linux server with 78GB of RAM.

5.1 Constructing worst-case terms for𝑚-CFA
In order to both evaluate the performance and verify the correctness of our implementation, we
construct a family of terms that incur maximum-possible work in practice. Van Horn’s dissertation
details a construction of terms whose analysis require exponential work for 𝑘-CFA [18]. Figure 11
shows an example for 1-CFA. The ultimate issue is that #t and #f will be conflated through the call
to f, producing two runtime closures but four abstract closures; more bindings, or arguments to𝑤 ,
may be added to add further complexity.

We use Van Horn’s technique to generate high-complexity terms for our𝑚-CFA implementation.
The key trick to fool [k=1]-CFA from Van Horn’s example is the application of the identity function

((lambda (f)
(let ((m (f #t))

(n (f #f)))
m))

(lambda (z)
((lambda (x) x)
(lambda (w) (w z z)))))

Fig. 11. Example (from Van Horn) showing exponential behavior for k-CFA

((lambda (f)
(let ((mm (f M)

...
(m1 (f 1))
(n0 (f 0)))

m))
(lambda (z)
((lambda (x) (+ z (+ z ...)))
(lambda (x) x)))))

Fig. 12. An example term from our experiments.

Table 1. Running time and memory usage of𝑚-CFA in Datalog. Term size N/K means N calls to f and K
invocations of +.

0 𝑝𝑎𝑑𝑑𝑖𝑛𝑔 1 𝑝𝑎𝑑𝑑𝑖𝑛𝑔 2 𝑝𝑎𝑑𝑑𝑖𝑛𝑔
Term size Polyvariance (𝑚) Time Memory Time Memory Time Memeory

0 00:09:57 1.27GB 00:09:46 1.86GB 09:48.04 1.27GB
32/4 1 < 1 sec 12.1MB 00:22:49 1.28GB 13:26.88 1.27GB

2 < 1 sec 12.29MB < 1 sec 12.5MB 19:09.89 1.27GB

0 01:08:58 3.55GB 01:09:36 3.55GB 01:02:51 3.56GB
86/3 1 < 1 sec 6.31MB 01:34:37 3.55GB 01:32:10 3.56GB

2 < 1 sec 6.31MB < 1 sec 6.32MB 02:13:10 3.56GB

inside the term. This identity function acts as padding, as [k=1]-CFA traces only the most recent
call site. During concrete execution, the intermediate call to the identity function sits (on the stack)
in front of the (separate) calls to f. In [k=1]-CFA, only the most recent callsite is remembered
and thus #t and #f will flow to same address. However we cannot directly translatiterate this
example into𝑚-CFA as we will not see the expected storage blowup. In𝑚-CFA, the context (the
contour in 𝑘-CFA) only grows in the state when values are bound to a variable. If padding is in
function postion when _ (w) (w x) is evaluated, the context has not been extended yet. Using
Van Horn’s example with𝑚-CFA, the padding will be bypassed. In𝑚-CFA, the padding should be
moved into argument position and [-expand the precision losing term. This is shown in Figure
12. This modification forces conflation of the values after the padding label is appended into the
context.

Table 2. Parallel Performance of m-CFA Soufflé implementation. 32 different clauses in let with 2 padding

1 core 2 threads 4 threads 8 threads
𝑚 time memory time memory time memeory time memory

0 00:09:48 1.27GB 00:13:05 1.43GB 00:18:39 1.55GB 00:22:38 1.61GB
1 00:13:26 1.27GB 00:14:36 1.46GB 00:20:36 1.55GB 00:22:54 1.62GB
2 00:19:09 1.27GB 00:25:14 1.46GB 00:36:11 1.54GB 00:46:57 1.60GB

5.2 Results
We used our worst-case term construction to perform a variety of runs using Soufflé. Table 1 shows
the results of our system on a set of two terms: one with 32 let bindings and one with 86 let
bindings. We measure 0, 1, and 2-CFA using a variety of paddings (0, 1, and 2). By construction, we
expect terms to explode when the amount of padding is too low as the analysis begins to conflate
polynomially-greater callsites. For example, in 32/4, we see 0-CFA explode while 1 and 2-CFA are
very fast. This is because both 1 and 2-CFA are fully-precise for the term. As expected, analyses of
higher-complexity have longer runtimes. For example, looking at the timings for 2 padding, we can
see that 2-CFA is roughly twice as slow as 0-CFA.

We measured parallel performance using a variety of thread counts. Soufflé was built to support
scalability and multithreading to speed up parallel execution on a single node. In our analysis, we
believed there would be many potentially-parallelizable states. For example, we evaluate differ-
ent let clauses non-deterministically. We used Soufflé version 2.02, compiling from source with
OpenMP support enabled. Our results were all compiled to C++ via Soufflé’s -c flag. We verified
CPU utilization (via htop) to ensure multithreading was enabled. However, as detailed in Table 2,
our results demonstrated anti-scalability: instead of making the program run faster, the more cores
used, the slower execution we observed, while also adding more memory overhead. GitHub issues
from the Soufflé authors point to several potential reasons for poor scalability. For example, our
implementation uses a large number of rules, and Soufflé does not parallelize across rules.

6 CONCLUSION
Datalog-based implementation of static analysis tools has enabled new frontiers in the scalability of
analyses to object-oriented languages. However, we do not know of any presentations that extend
these ideas to Scheme-like languages. This paper presented the key ideas necessary to implement
analyses of Scheme-like languages using Datalog. We structure our analysis using the AAM-based
approach to facilitate translation to Datalog’s deductive rules, using the𝑚-CFA allocation strategy of
Might et al. to mirror the flat closure structure naturally enabled by Datalog [11]. To our knowledge,
this is the first presentation of a Datalog-based analysis for Scheme-like languages.

A FULL SOUFFLÉ IMPLEMENTATION
.type id <: symbol

.type context = Context{ctx0:id}

.type value = Number { n : number }
| Bool { b : symbol }
| Kont { k : address_k }
| Closure { e: id, ctx: context }
| PrimVal {op: id , v1: value , v2: value}

.type kont = MT {}
| Arg {args: id, ctx: context, ectx: context, next_ak: address_k}
| Fn {fn: value, pos: number, ctx: context, next_ak: address_k}
| Set {loc: address_v, next_ak: address_k}
| If {true_branch: id, false_branch: id, ctx: context, next_ak: address_k}
| Callcc {ectx: context, next_ak: address_k}
| Let {av: address_v, ebody: id, ctx: context, next_ak: address_k}
| Prim1 {op: id, e2: id, ctx: context, next_ak: address_k}
| Prim2 {op: id, v1: value, next_ak: address_k}

.type address_k = KAddress{e: id, ctx: context}

.type address_v = VAddress{x: symbol, ctx: context}

.decl state_e(e: id, ctx: context, ak: address_k)

.output state_e

.decl state_a(v: value, ak: address_k)

.output state_a

.decl stored_val(av: address_v, v: value)

.output stored_val

.decl stored_kont(ak: address_k, k: kont)

.output stored_kont

.decl lambda(Id: id, Vars: id, BodyId: id)

.input lambda

.decl lambda_arg_list(Id: id, Pos: number, X: symbol)

.input lambda_arg_list

.decl prim(Id: id, OpName: symbol)

.input prim

.decl prim_call(Id: id, PrimId: id, Args: id)

.input prim_call

.decl call(Id: id, FuncId: id, Args: id)

.input call

.decl call_arg_list(Id: id, Pos: number, X: id)

.input call_arg_list

.decl var(Id: id, MetaName: symbol)

.input var

.decl num(Id: id, v: number)

.input num

.decl bool(Id: id, v: symbol)

.input bool

.decl quotation(Id: id, Expr: id)

.input quotation

.decl value_form(Id: id)
value_form(id) :-

(num(id, _); var(id, _); lambda(id, _, _); quotation(id, _); bool(id, _)).

.decl if(Id:id, GuardId: id, TrueId: id, False: id)

.input if

.decl setb(Id: id, Var: symbol, ExprId: id)

.input setb

.decl callcc(Id: id, ExprId: id)

.input callcc

.decl let(Id: id, BindId: id, BodyId: id)

.input let

.decl let_list(Id: id, X: symbol, EId: id)

.input let_list

.decl top_exp(Id: id)

.input top_exp

.decl freevar(x:symbol, e: id)

.output freevar

freevar(x, e) :- var(e, x).
freevar(x, e) :-

lambda(e, vars, body),
freevar(x, body),
!lambda_arg_list(vars, _, x).

freevar(x, e) :-
call(e, func, args),
(freevar(x, func); freevar(x, args)).

freevar(x, e) :-
prim_call(e, _, args),
freevar(x, args).

freevar(x, e) :-
call_arg_list(e, pos, arg),
freevar(x, arg).

freevar(x, e) :-
if(e, eguard, et, ef),
(freevar(x, eguard); freevar(x, et); freevar(x, ef)).

freevar(y, e) :-
setb(e, _, ev),
freevar(y, ev).

freevar(x, e) :-
callcc(e, ev),
freevar(x, ev).

freevar(x, e) :-
let(e, binds, body),
(freevar(x, binds); freevar(x, body)).

freevar(x, e) :-
let_list(e, a, bind),
!let_list(e, x, _),
freevar(x, bind).

.decl flow_ee(e1: id, e2: id)

.output flow_ee

.decl flow_ea(e1: id, a2: value)

.output flow_ea

.decl flow_aa(a1: value, a2: value)

.output flow_aa

.decl flow_ae(a1: value, e2: id)

.output flow_ae

.decl peek_ctx(e: id, ctx_old: context, ctx_new: context)

.output peek_ctx

.decl copy_ctx(from: context, to: context, e:id)

.output copy_ctx

state_e(e, $Context(""), $KAddress(e, $Context(""))),
peek_ctx(e, $Context(""), $Context(e)),
stored_kont($KAddress(e, $Context("")), $MT) :-

top_exp(e).

peek_ctx(e, $Context(ctx0), $Context(e)) :-
state_e(e, $Context(ctx0), _),
(callcc(e, _) ; call(e, _, _)
; let(e, _, _); lambda(e,_,_)).

stored_val($VAddress(fv, to), v) :-
copy_ctx(from, to, e),
freevar(fv, e),
stored_val($VAddress(fv, from), v).

state_e(eguard, ctx, $KAddress(eguard, ctx)),
stored_kont($KAddress(eguard, ctx), $If(et, ef, ctx, ak)),
flow_ee(e, eguard) :-

state_e(e, ctx, ak),
if(e, eguard, et, ef).

state_e(elam, ctx, $KAddress(elam, ctx)),
stored_kont($KAddress(elam, ctx), $Callcc(ectx, ak)),
flow_ee(e, elam) :-

state_e(e, ctx, ak),
callcc(e, elam),
peek_ctx(e, ctx, ectx).

state_e(esetto, ctx, $KAddress(esetto, ctx)),
stored_kont($KAddress(esetto, ctx), $Set($VAddress(x, ctx), ak)),
flow_ee(e, esetto) :-

state_e(e, ctx, ak),
setb(e, x, esetto).

state_e(efunc, ctx, $KAddress(efunc, ctx)),
stored_kont($KAddress(efunc, ctx), $Arg(eargs, ctx, ectx, ak)),
flow_ee(e, efunc) :-

state_e(e, ctx, ak),
call(e, efunc, eargs),
peek_ctx(e, ctx, ectx).

state_e(ebnd, ctx, $KAddress(ebnd, ctx)),
stored_kont($KAddress(ebnd, ctx), $Let($VAddress(x, ectx), ebody, ectx, ak)),
copy_ctx(ctx, ectx, e),
flow_ee(e, ebnd) :-

state_e(e, ctx, ak),
let(e, ll, ebody),
let_list(ll, x, ebnd),
peek_ctx(e, ctx, ectx).

state_e(earg0, ctx, $KAddress(earg0, ctx)),
stored_kont($KAddress(earg0, ctx), $Prim1(op, earg1, ctx, ak)),
flow_ee(e, earg0) :-

state_e(e, ctx, ak),
prim_call(e, op, pl),
call_arg_list(pl, 0, earg0),
call_arg_list(pl, 1, earg1).

state_a($Number(n), ak),
flow_ea(e, $Number(n)) :-

state_e(e, ctx, ak),
num(e, n).

state_a($Bool(b), ak),
flow_ea(e, $Bool(b)) :-

state_e(e, ctx, ak),
bool(e, b).

state_a($Closure(e, ctx), ak),
flow_ea(e, $Closure(e, ctx)) :-

state_e(e, ctx, ak),
lambda(e, _, _).

state_a(v, ak),
flow_ea(e, v) :-

state_e(e, ctx, ak),
var(e, x),
stored_val($VAddress(x, ctx), v).

state_e(et, ctx_k, next_ak),
flow_ae($Bool("#t"), et) :-

(state_a($Bool("#t"), ak) ; state_a($Closure(_,_), ak)
; state_a($Number(_), ak) ; state_a($Kont(_), ak)),

stored_kont(ak, $If(et, _, ctx_k, next_ak)).

state_e(ef, ctx_k, next_ak),
flow_ae($Bool("#f"), ef) :-

state_a($Bool("#f"), ak),
stored_kont(ak, $If(_, ef, ctx_k, next_ak)).

state_e(ebody, ectx, next_ak),
stored_val($VAddress(x, ectx), $Kont(ak)),
copy_ctx(ctx_clo, ectx, elam),
flow_ae($Closure(elam, ctx_clo), ebody) :-

state_a($Closure(elam, ctx_clo), ak),
stored_kont(ak, $Callcc(ectx, next_ak)),
lambda(elam, params, ebody),
lambda_arg_list(params, 0, x).

state_a($Kont(ak), bk),
flow_aa($Kont(bk), $Kont(ak)) :-

state_a($Kont(bk), ak),
stored_kont(ak, $Callcc(_, _)).

state_e(earg, ctx, $KAddress(earg, ctx)),
stored_kont($KAddress(earg, ctx), $Fn(v, pos, ectx, next_ak)),
flow_ae(v, earg) :-

state_a(v, ak),
stored_kont(ak, $Arg(eargs, ctx, ectx, next_ak)),
call_arg_list(eargs, pos, earg).

state_e(ebody, ectx, next_ak),
stored_val($VAddress(x, ectx), v),
copy_ctx(ctx_clo, ectx, elam),
flow_ae(v, ebody) :-

state_a(v, ak),
stored_kont(ak, $Fn($Closure(elam, ctx_clo), pos, ectx, next_ak)),
lambda(elam, params, ebody),

lambda_arg_list(params, pos, x).

state_a(v, callcc_kont),
flow_aa(v, v) :-

state_a(v, ak),
stored_kont(ak, $Fn($Kont(callcc_kont), 0, _, _)).

state_e(ebody, ctx, next_ak),
stored_val(av, v),
flow_ae(v, ebody) :-

state_a(v, ak),
stored_kont(ak, $Let(av, ebody, ctx, next_ak)).

state_e(earg1, ctx, $KAddress(earg1, ctx)),
stored_kont($KAddress(earg1, ctx), $Prim2(op, v, next_ak)),
flow_ae(v, earg1) :-

state_a(v, ak),
stored_kont(ak, $Prim1(op, earg1, ctx, next_ak)).

state_a($PrimVal(op, v1, v2), next_ak),
flow_aa(v2, $PrimVal(op, v1, v2)) :-

state_a(v2, ak),
stored_kont(ak, $Prim2(op, v1, next_ak)).

state_a($Number(-42), next_ak),
stored_val(loc, v),
flow_aa(v, $Number(-42)) :-

state_a(v, ak),
stored_kont(ak, $Set(loc, next_ak)).

